NAG Library Routine Document

F07WWF (ZPFTRI)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07WWF (ZPFTRI) computes the inverse of a complex Hermitian positive definite matrix using the Cholesky factorization computed by F07WRF (ZPFTRF) stored in Rectangular Full Packed (RFP) format.

2 Specification

```
SUBROUTINE F07WWF (TRANSR, UPLO, N, AR, INFO)
INTEGER                 N, INFO
COMPLEX (KIND=nag_wp) AR(N*(N+1)/2)
CHARACTER(1)            TRANSR, UPLO
```

The routine may be called by its LAPACK name ***zpftri***.

3 Description

F07WWF (ZPFTRI) is used to compute the inverse of a complex Hermitian positive definite matrix A, stored in RFP format. The RFP storage format is described in Section 3.3.3 in the F07 Chapter Introduction. The routine must be preceded by a call to F07WRF (ZPFTRF), which computes the Cholesky factorization of A.

If UPLO = 'U', $A = U^{\mathrm{H}}U$ and A^{-1} is computed by first inverting U and then forming $(U^{-1})U^{-\mathrm{H}}$.

If UPLO = 'L', $A = LL^{\mathrm{H}}$ and A^{-1} is computed by first inverting L and then forming $L^{-\mathrm{H}}(L^{-1})$.

4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

Gustavson F G, Waśniewski J, Dongarra J J and Langou J (2010) Rectangular full packed format for Cholesky's algorithm: factorization, solution, and inversion *ACM Trans. Math. Software* **37, 2**

5 Arguments

1: TRANSR – CHARACTER(1) *Input*

On entry: specifies whether the normal RFP representation of A or its conjugate transpose is stored.

TRANSR = 'N'
The matrix A is stored in normal RFP format.

TRANSR = 'C'
The conjugate transpose of the RFP representation of the matrix A is stored.

Constraint: TRANSR = 'N' or 'C'.

2: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{H}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{H}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: AR(N $\times$ (N + 1)/2) – COMPLEX (KIND=nag_wp) array *Input/Output*

On entry: the Cholesky factorization of A stored in RFP format, as returned by F07WRF (ZPFTRF).

On exit: the factorization is overwritten by the n by n matrix A^{-1} stored in RFP format.

5: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The leading minor of order $\langle value \rangle$ is not positive definite and the factorization could not be completed. Hence A itself is not positive definite. This may indicate an error in forming the matrix A. There is no routine specifically designed to invert a Hermitian matrix stored in RFP format which is not positive definite; the matrix must be treated as a full Hermitian matrix, by calling F07MWF (ZHETRI).

7 Accuracy

The computed inverse X satisfies

$$\|XA - I\|_2 \leq c(n)\epsilon\kappa_2(A) \quad \text{and} \quad \|AX - I\|_2 \leq c(n)\epsilon\kappa_2(A),$$

where $c(n)$ is a modest function of n, ϵ is the ***machine precision*** and $\kappa_2(A)$ is the condition number of A defined by

$$\kappa_2(A) = \|A\|_2\left\|A^{-1}\right\|_2.$$

8 Parallelism and Performance

F07WWF (ZPFTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of real floating-point operations is approximately $\frac{8}{3}n^3$.

The real analogue of this routine is F07WJF (DPFTRI).

10 Example

This example computes the inverse of the matrix A, where

$$A = \begin{pmatrix} 3.23 + 0.00i & 1.51 - 1.92i & 1.90 + 0.84i & 0.42 + 2.50i \\ 1.51 + 1.92i & 3.58 + 0.00i & -0.23 + 1.11i & -1.18 + 1.37i \\ 1.90 - 0.84i & -0.23 - 1.11i & 4.09 + 0.00i & 2.33 - 0.14i \\ 0.42 - 2.50i & -1.18 - 1.37i & 2.33 + 0.14i & 4.29 + 0.00i \end{pmatrix}.$$

Here A is Hermitian positive definite, stored in RFP format, and must first be factorized by F07WRF (ZPFTRF).

10.1 Program Text

```
    Program f07wwfe

!     F07WWF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dbf, zpftrf, zpftri, ztfttr
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, k, lar1, lda, lenar, &
                                          n, q
      Character (1)                    :: transr, uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:), ar(:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Executable Statements ..
      Write (nout,*) 'F07WWF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, uplo, transr

      lenar = n*(n+1)/2
      lda = n
      Allocate (ar(lenar),a(lda,n))

!     Setup notional dimensions of RFP matrix AR
      k = n/2
      q = n - k
      If (transr=='N' .Or. transr=='n') Then
        lar1 = 2*k + 1
      Else
        lar1 = q
      End If

!     Read an RFP matrix into array AR
      Do i = 1, lar1
        Read (nin,*) ar(i:lenar:lar1)
      End Do

!     Factorize A
!     The NAG name equivalent of zpftrf is f07wrf
      Call zpftrf(transr,uplo,n,ar,info)

      Write (nout,*)
```

```
      Flush (nout)
      If (info==0) Then

!       Compute inverse of A
!       The NAG name equivalent of zpftri is f07wwf
        Call zpftri(transr,uplo,n,ar,info)

!       Convert and print inverse
!       The NAG name equivalent of ztfttr is f01vhf
        Call ztfttr(transr,uplo,n,ar,a,lda,info)
        ifail = 0
        Call x04dbf(uplo,'Nonunit',n,n,a,lda,'Bracketed','F7.4','Inverse',     &
          'Integer',rlabs,'Integer',clabs,80,0,ifail)

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07wwfe
```

10.2 Program Data

```
F07WWF Example Program Data
   4     'L'    'N'           : n, uplo, transr
 ( 4.09, 0.00)  ( 2.33,-0.14)
 ( 3.23, 0.00)  ( 4.29, 0.00)
 ( 1.51, 1.92)  ( 3.58, 0.00)
 ( 1.90,-0.84)  (-0.23,-1.11)
 ( 0.42,-2.50)  (-1.18,-1.37) : AR
```

10.3 Program Results

```
 F07WWF Example Program Results

 Inverse
                     1                 2                 3                 4
 1  ( 5.4691, 0.0000)
 2  (-1.2624,-1.5491) ( 1.1024, 0.0000)
 3  (-2.9746,-0.9616) ( 0.8989,-0.5672) ( 2.1589,-0.0000)
 4  ( 1.1962, 2.9772) (-0.9826,-0.2566) (-1.3756,-1.4550) ( 2.2934, 0.0000)
```